
Abstract. In this paper, we present techniques for automated verifica-
tion of behavioural specifications using hidden algebra. Two non-trivial
examples, the Alternating Bit Protocol and a snooping cache coher-
ence protocol, are presented with complete specification code and proof
scores for CafeOBJ verification system. The refinement proof based on
behavioural coinduction is given for the first example, and the coherence
proof based on invariance is given for the second.

1 Introduction

The promise of formal methods has been heard for a long time, but with a few
exceptions, they have not lived up to expectations. The shortcomings are due
to:

– gap between specification and implementation, leading to the overspecifica-
tion problem, and

– lack of unified logical systems that cover entire software design/development
process, necessitating human intervention.

In other words, there have not been appropriate logical foundations for software
specification. Overpopulation of specification languages has led to idiosyncrasy
that is so difficult to see through. If formal methods are to have industrial im-
pact on software productivity, they have to offer seamless integrated support
throughout specification, verification, and transformation.

In this paper, we report on several techniques developed for behavioural
specification based on hidden algebra to demonstrate its potential to be an
industrial-strength specification method in the future. Hidden algebra was de-
veloped in an attempt to give a semantics for software engineering, and for the
object paradigm in particular, supporting correctness proofs that are as simple
and mechanical as possible [1]. It distinguishes hidden sorts from visible sorts.
As hidden sorts model states of the system, equations of hidden sorts need to be
interpreted in a different manner. For this, hidden algebra formalises a notion
of behavioural abstraction, by defining behavioural satisfaction of equations.
An effective proof method for behavioural equation has been developed, called
behavioural coinduction, and used for various refinement proofs.

The development of hidden algebra is not an isolated event. There are many
other proposed methods for behavioural specification, whose purpose is to char-
acterise how systems behave instead of how they are implemented, but very few
computer support systems exist. The CafeOBJ [2] system offers an integrated
environment for specification and verification in the tradition of the OBJ lan-
guages, and is the only system that supports behavioural specification based
on hidden algebra. As part of an effort to show the capability of behavioural
specification, we report on some of the new features of CafeOBJ by means of
two non-trivial examples, the Alternating Bit Protocol and a snooping cache
coherence protocol. Specification code and correctness proof scores for CafeOBJ
are presented for a complete explanation.

The organisation of the paper is as follows: Section 2 briefly summarises the-
oretical backgrounds of hidden algebra, Section 3 and 4 detail the specification
and verification of the Alternating Bit Protocol and snooping cache coherence
protocol respectively, and Section 5 concludes with comparisons with other ap-
proaches and future plans.

2 Hidden Algebra

This section presents basic definitions of hidden algebra. See [1, 3] for more
details.

Hidden algebra distinguishes hidden state values from visible data values.
Data values are defined as elements of a fixed (ordinary) algebraD with signature
Ψ and sort set V such that for each d ∈ Dv with v ∈ V there is some ψ ∈ Ψ[],v

interpreted as d in D. On the other hand, (hidden) state values are defined as
elements of a special algebra with syntactic restrictions on its signature.

Definition 1. A hidden signature (over (V, Ψ,D)) is a triple (H,Σ,Σb),
where H is a set of hidden sorts disjoint from V , Σ is an (H ∪ V )-sorted
signature with Ψ ⊆ Σ, and Σb ⊆ Σ is a set of behavioural operations disjoint
from Ψ , such that

– each σ ∈ Σw,s with w ∈ V ∗ and s ∈ V lies in Ψw,s, and,
– each σ ∈ Σb

w,s has exactly one element of H in w.

(H,Σ,Σb) may be abbreviated to Σ leaving Σb implicit. An operation σ in
Σb

w,s is called a method if s ∈ H and an attribute if s ∈ V . σ ∈ Σw,s is called
a hidden constant if w ∈ V ∗ and s ∈ H.

Note that operations in Σw,s−Σb
w,s may have more than one elements of H in w.

These non-behavioural operations were not considered in the original defini-
tion of hidden algebra [1]. They have been introduced to increase expressiveness
of hidden algebra [4, 5]. The above definition is due to [4] and is supported in
the current CafeOBJ system.

Definition 2. Given a hidden signature (H,Σ,Σb), a hidden Σ-algebra A is
a (many sorted) Σ-algebra A such that A|̀Ψ = D.

As mentioned above, the elements of Av where v ∈ V is thought of as data values
and the elements of Ah where h ∈ H as state values.

Definition 3. A hidden (or behavioural) theory (or specification) is a
quadruple (H,Σ,Σb, E), where (H,Σ,Σb) is a hidden signature and E is a set
of Σ-(conditional) equations; we may write (Σ,E) for short.

Example 1. We present a behavioural specification of a flag object using CafeOBJ
notations [3]. A flag object is either up or down, and there are methods to put
it up, to put it down, and to reverse its state:

mod* FLAG {
*[ Flag ]*
bops (up_) (dn_) (rev_) : Flag -> Flag -- methods
bop up?_ : Flag -> Bool -- attribute
var F : Flag
eq up? up F = true .
eq up? dn F = false .
eq up? rev F = not up? F .

}

A CafeOBJ keyword mod* means that the module has a loose behavioural se-
mantics in contrast with a tight (initial algebra) semantics specified by the
keyword mod!. A pair of starred brackets *[ ... ]* is used for sort declaration
and * indicates that the declared sort is a hidden sort. For visible sorts, [ ... ]
is used. The keyword bop declares behavioural operations in Σb (i.e., attributes
and methods).

The meaning of the FLAG specification should be clear, however, some of
the intended behaviours of the flag object, for example, a behavioural equation
rev rev F = F cannot be deduced from the FLAG specification using ordinary
equational reasoning1. This means that ordinary satisfaction of equations is too
strict for behavioural equations and a weaker notion of satisfaction based on
indistinguishability2 is needed. In hidden algebra, it is formalised using contexts
as follows.

Definition 4. Given a hidden signature (H,Σ,Σb), a behavioural context is
a term having a single occurrence of a special variable of hidden-sort denoted by
z and is formed by the following rules:

– any variable z (of any sort) is a behavioural context,
– for any σ ∈ Σb

vh,s, any behavioural context c of sort h, and any tuple of
ground terms t ∈ (TΣ)v, σ(t, c) is a behavioural context, where v ∈ V ∗ and
h ∈ H. If s ∈ V , then σ(t, c) is called visible, otherwise it is called hidden.

1 Not to mention induction.
2 By means of method application and attribute observation.

Given a hidden Σ-algebra A, two elements a and a′ of the same carrier set
As are said to be behaviourally equivalent, denoted by a ∼s a

′ (or just a ∼ a′)
iff Ac(a) = Ac(a′) for all visible behavioural contexts c, where Ac denotes the
function interpreting the context c as an operation on A.

A hidden Σ-algebra A behaviourally satisfies a (conditional) equation e
of the form (∀X) t = t′ if t1 = t′1, ..., tm = t′m iff

θ∗(t) ∼ θ∗(t′) whenever θ∗(tj) ∼ θ∗(t′j) for all j = 1, ...,m

for every valuation θ : X → A. In this case, we write A |≡Σ e. We may drop the
subscript Σ.

Note that for visible sorted equations, there is no difference between ordinary
satisfaction and behavioural satisfaction. We will use the symbol ∼ instead of
= when the equation should be interpreted by behavioural satisfaction. Such
equations are called behavioural equations and specified by keywords beq,
bceq in CafeOBJ, instead of eq, ceq where ceq stands for conditional equations.

The first effective algebraic proof technique for behavioural equivalence was
context induction [6], however, a more comprehensive technique based on max-
imality has been developed.

Definition 5. Given a hidden signature Σ, and a hidden Σ-algebra A, be-
havioural congruence on A is a Σb-congruence which is identity on visible
sorts.

Theorem 1. Given a hidden signature Σ and a hidden Σ-algebra A, then be-
havioural equivalence is the largest behavioural congruence on A.

See [1, 4, 5] for the proof.
Thanks to this theorem, one can show a ∼ a′ by finding some behavioural

congruence that relates a and a′ 3. This is what is called behavioural (or hid-
den) coinduction and justifies a variety of techniques for proving behavioural
equivalence. For example, to show that every FLAG-algebra satisfies the equation
(∀F : Flag) rev rev F = F, one only needs to show that up? rev rev F = up? F.
This is a special case of attribute coherent theory, where the equivalence on
attributes is behavioural equivalence 4. CafeOBJ system automatically checks if
this happens every time a new module is loaded. When this is the case, showing
behavioural satisfaction is automatic.

Non-behavioural operations have been introduced to enhance the expres-
sive power of hidden algebra [4], for example for modular construction of be-
havioural specifications. However, non-behavioural constructors may not pre-
serve behavioural equivalence and therefore may ruin the soundness of equa-
tional reasoning (specifically the congruence rule substituting equal with equal).
The next definition gives a sufficient condition for sound equational deduction.
3 Note that the situation is very much similar to the technique in process algebra for

demonstrating (strong) bisimilarity through bisimulation [7].
4 Visit the website at UCSD, http://www.cs.ucsd.edu/groups/links.

Definition 6. Given a hidden signature Σ and a hidden Σ-algebra A, an oper-
ation σ ∈ Σw,s−Σb

w,s is said to be behaviourally coherent iff it preserves the
behavioural equivalence on A, that is, Aσ(a1, a2, . . . , an) ∼s Aσ(a′1, a

′
2, . . . , a

′
n) if

a1 ∼s1 a
′
1, a2 ∼s2 a

′
2, . . . , an ∼sn a′n for all (a1, a2, . . . , an) and (a′1, a

′
2, . . . , a

′
n)

in As1 × As2 × · · · ×Asn , where w = s1s2 · · · sn.

The next theorem is due to [4]. See also [5] for another proof (in a slightly
different setting) and examples.

Theorem 2. If all operations in Σ −Σb are behaviourally coherent, then ordi-
nary equational deduction is sound for behavioural equations.

3 The Alternating Bit Protocol

Now that we have introduced all necessary concepts, let us turn to some inter-
esting specifications and their verification.

The first example is the classic Alternating Bit Protocol (ABP) [8]. The pro-
tocol is designed to achieve secure communication through unreliable channels
that may lose or duplicate packets. The model of the protocol consists of four
agents, the sender, the receiver, the message (msg) channel from the sender to
the receiver, and the acknowledgement (ack) channel from the receiver to the
sender.

Sender  ReceiverChannels

.........

.........

.........  ack

data
 bit

 msg

 ack

Fig. 1. The Alternating Bit Protocol

In addition to data that are transmitted, the protocol uses extra bits to
recover potential errors (loss and duplication) in the channels. The sender sends
messages, each of which consists of a data and a bit, along the msg channel,
and receives acknowledgement bits from the ack channel. The receiver receives
messages from the msg channel and sends back acknowledgements along the
ack channel. The channel is modeled as a simple unbounded queue that may
lose or duplicate packets, but does not alter the order of packets. As its name
suggests, the key trick of the protocol is the bit alternation that takes place
when a packet having the right bit is received. This is a good example for formal

specification and verification methods, since its procedure is extremely simple,
yet very sophisticated.

We model the protocol in terms of actions taken by the sender and the
receiver. Each agent maintains three types of information, a data, a bit, and a
timer. Each checks the incoming bit with the bit it has and then determines
the next action. We have included the timer since it is necessary to prevent
deadlocks5.

Sender:

– receives a right ack (the same bit as it has) from the ack channel, accepts
and keeps a new data, alternates the bit, sends a couple of the new data and
the new bit along the msg channel, sets a timer;

– receives a wrong ack (does nothing);
– resends a couple of the current data and bit when the timer goes off.

Receiver:

– receives a message with a right bit (the opposite bit of the one it has) from
the msg channel, keeps the new data (for future delivery), alternates the bit,
sends the new bit along the ack channel, sets a timer;

– receives a message with a wrong bit (does nothing);
– resends the current bit when the timer goes off.

Below is a more or less faithful behavioural specification of the descriptions
above.

mod! DATA { protecting(BOOL) [ Nat Bool < Data ] }

mod! QUEUE(X :: TRIV) {
[ NeQueue < Queue ]
op nil : -> Queue
op front : NeQueue -> Elt
op enq : Elt Queue -> NeQueue
op deq : NeQueue -> Queue
vars D E : Elt var Q : Queue
eq deq(enq(E,nil)) = nil .
eq deq(enq(E,enq(D,Q))) = enq(E,deq(enq(D,Q))) .
eq front(enq(E,nil)) = E .
eq front(enq(E,enq(D,Q))) = front(enq(D,Q)) .

}
mod* SENDER {
protecting(DATA)
*[ Sender ]*
bop bit : Sender -> Bool
bop val : Sender -> Data

5 However, the timer does not appear in the behavioural specification.

bop in : Data Bool Sender -> Sender
op init : -> Sender
var D : Data var B : Bool var S : Sender
eq bit(init) = true . -- valid initial state
ceq val(in(D,B,S)) = D if bit(S) == B.--new data for right ack
ceq bit(in(D,B,S)) = not bit(S) if bit(S) == B.--alternates bit
bceq in(D,B,S) = S if bit(S) =/= B . -- stays put for wrong ack

}
mod* RECEIVER {
protecting(DATA)
*[ Receiver ]*
bop bit : Receiver -> Bool
bop val : Receiver -> Data
bop get : Data Bool Receiver -> Receiver
op init : -> Receiver
var D : Data var B : Bool var R : Receiver
eq bit(init) = true . -- valid initial state
ceq val(get(D,B,R)) = D if bit(R) =/= B . -- output value
ceq bit(get(D,B,R)) = not bit(R) if bit(R)=/=B.--alternates bit
bceq get(D,B,R) = R if bit(R) == B . -- stays put for wrong bit

}
mod* ABP {
protecting(SENDER + RECEIVER + QUEUE[DATA])
*[ Abp ]*
op Init : -> Abp
op Protocol: Sender Receiver Queue Queue Queue->Abp {coherent}
bop In : Data Abp -> Abp
bop Out : Abp -> Abp
bop Val : Abp -> Data
vars D E : Data var B : Bool var A : Abp var S : Sender
var R : Receiver vars L L1 L2 : Queue
beq Init = Protocol(init,init,nil,nil,nil) .
bceq In(D,Protocol(S,R,L1,L2,enq(B,L)))

= Protocol(in(D,front(enq(B,L)),S),R,enq(D,L1),
enq(not bit(S),L2),deq(enq(B,L)))

if bit(S) == front(enq(B,L)) .
beq In(D,Protocol(S,R,enq(E,L1),enq(B,L2),nil))

= Protocol(S,R,enq(E,L1),enq(B,L2),nil) .
bceq [ 1 ] : Protocol(S,R,L1,L2,enq(B,L))

= Protocol(S,R,L1,L2,deq(enq(B,L)))
if bit(S) =/= front(enq(B,L)) .

bceq Out(Protocol(S,R,enq(D,L1),enq(B,L2),L))
= Protocol(S,get(front(enq(D,L1)),front(enq(B,L2)),R),
deq(enq(D,L1)),deq(enq(B,L2)),enq(not bit(R),L))

if bit(R) =/= front(enq(B,L2)) .

bceq [ 2 ] : Protocol(S,R,enq(D,L1),enq(B,L2),L)
= Protocol(S,R,deq(enq(D,L1)),deq(enq(B,L2)),L)
if bit(R) == front(enq(B,L2)) .

beq Out(Protocol(S,R,nil,nil,enq(B,L)))
= Protocol(S,R,nil,nil,enq(B,L)) .

beq [ 3 ] : Protocol(S,R,L1,L2,L)
= Protocol(S,R,enq(val(S),L1),enq(bit(S),L2),L) .

beq [ 4 ] : Protocol(S,R,L1,L2,L)
= Protocol(S,R,L1,L2,enq(bit(R),L)) .

eq Val(Protocol(S,R,L1,L2,L)) = val(R) .
}

A few remarks about the specification are in order.

– Transmitted data and alternation bits are specified using CafeOBJ’s default
built-in modules Nat (the natural numbers) and Bool (the Booleans).

– A parameterised module QUEUE and subsorts Nat and Bool are used to avoid
multiple presence of QUEUE modules. Also subsort NeQueue is used to avoid
introducing user-defined error elements.

– Modules are imported by protecting declaration. It instructs that the se-
mantics of imported modules should not be altered. There are other impor-
tation modes than protecting. See [3] for details and semantics.

– Initial states are defined as hidden constants. Attribute values for initial
states can be defined by visible equations. The initial state of ABP may look
strange since the data transmission can only be started by the Receiver
taking a time-out action6. This is rather a syntactic convenience, i.e., one
could have put an appropriate acknowledgement bit in the ack channel.

– Module expressions such as SENDER + RECEIVER + QUEUE[DATA] avoid mul-
tiple internal copies of shared modules.

– A non-behavioural operation Protocol is declared coherent in the ABP
module. Thus one can freely use equational reasoning through reduction
commands reduce or red in CafeOBJ. The coherence of Protocol needs to
be proved separately, however, it is trivial and omitted here. One can think
of coherent operations as behavioural constructors just like ordinary data
type constructors. They are used to define legitimate system configurations
in terms of behavioural sub-modules.

CafeOBJ keywords appearing in the specification should be easily interpreted
by corresponding hidden algebra notions. Interested readers are referred to [3]
for CafeOBJ specific details.

Interpreting equations [ 1 ], [ 2 ], [ 3 ], and [ 4 ] with (ordinary) strict
satisfaction of equations will not make sense. For example, one might think
that equations [ 3 ] and [ 4 ] bring about inconsistency since the operation
Protocol acts as a constructor. One has to remember, however, that they are
behavioural equations that only need to be satisfied behaviourally. By virtue of

6 Not by the Sender.

behavioural satisfaction, equations [ 1 ] and [ 2 ] state that wrong packets
are removed without making any trace while [ 3 ] and [ 4 ] state that time-
out events are invisible and resending actions should not affect the following
behaviour of the protocol. Writing behavioural equations always runs a risk of
producing inconsistent specifications since they imply infinite number of visible
equations. This is probably the most difficult aspect of behavioural specification.

We define the notion of behavioural refinement first.

Definition 7. A hidden signature map ϕ : (H,Σ) → (H ′, Σ′) is a signature
morphism ϕ : Σ → Σ′ that preserves hidden sorts and behavioural operations,
and that is the identity on (V, Ψ). A hidden signature map ϕ : (Σ,E) → (Σ′, E′)
is a refinement iff for every (Σ′, E′)-algebra M ′ we have ϕM ′ |≡Σ E. (ϕM ′

denotes M ′ viewed as a Σ-algebra.)

It can be shown that ϕ is a refinement iff all visible consequences of the abstract
specification hold in the concrete specification [9]:

Proposition 1. A hidden signature map ϕ : (Σ,E) → (Σ′, E′) is a refinement
iff E′ |= ϕ(c[e]) for each e ∈ E and each visible Σ-context c, where if e is the
equation (∀X) t = t′, then c[e] denotes the equation (∀X) c[t] = c[t′].

Now we want to show that ABP is a refinement of the following behavioural
specification BUF of a buffer of capacity one.

mod* BUF {
[ Nat ] *[ Buf ]*
op init : -> Buf
bop in : Nat Buf -> Buf
bop val : Buf -> Nat
bop out : Buf -> Buf
bop empty? : Buf -> Bool
var N : Nat var B : Buf
eq empty?(init) = true .
ceq empty?(out(B)) = true if not empty?(B) .
eq empty?(in(N,B)) = false .
ceq val(out(in(N,B))) = N if empty?(B) .
bceq in(N,B) = B if not empty?(B) .
bceq out(B) = B if empty?(B) .

}

The meaning of this specification should be clear, however, note that the output
value is secure only after the out operation.

In order to demonstrate7 refinement from BUF to ABP, we first need to find
an underlying signature map. For this, we define an ABP counterpart Empty? of
empty? as follows.

7 Before starting formal verification, one can check how ABP works via symbolic exe-
cution. See Appendix A for a CafeOBJ session of simple ABP reductions.

bop Empty? : Abp -> Bool .
var S : Sender var R : Receiver vars L L1 L2 : Queue
eq Empty?(Protocol(S,R,L1,L2,L)) = bit(S) == bit(R) .

Then the refinement proof requires us to show all equations from BUF are
valid in ABP as well. This is done by finding an appropriate candidate relation
R on Abp and showing that it is a behavioural congruence. In general, this is
highly non-trivial. However, a closer look at behavioural equations [ 1 ] – [ 4
] of ABP leads to the observation that for any reachable state of the protocol
there is a behaviourally equivalent state whose channels are all empty. This is
the ingenious aspect of the protocol that achieves robust communication against
loss and duplication in the channels.

We are going to make this observation precise by creating CafeOBJ scores.
However, since CafeOBJ does not have full theorem proving capability (only re-
ductions), we need some maneuvers for quantified variables and conditionals, and
also for rewriting directions and orders. The CafeOBJ commands for controlling
these are omitted from the following proof score.

We define the candidate relation R to be

op _R_ : Abp Abp -> Bool .
vars S1 S2 : Sender vars R1 R2 : Receiver
ceq Protocol(S1,R1,nil,nil,nil) R Protocol(S2,R2,nil,nil,nil)=true

if val(S1) == val(S2) and val(R1) == val(R2) and
((bit(S1) == bit(R1)) == (bit(S2) == bit(R2))) .

It suffices to define R on states with empty channels since behavioural equations
in ABP are closed within reachable states Protocol(S,R,L1,L2,L) where

L2 = a∗b∗, L = b∗a∗, a = bit(S), b = bit(R).

(The head of the queue is placed to the right, and a stands for the opposite
bit of a.) It is very interesting to note that only one bit change occurs in the
concatenation of L2 and L, depending upon whether a = b or a 6= b. The a = b
case corresponds to the empty state ready to accept a new data, and a 6= b
to the full state having an accepted data waiting to be delivered. Identifying
these two groups of states is essential for any attempt at formal verification of
the ABP. It is also noted that the use of a coherent non-behavioural operation
(that is, Protocol) simplifies the specification together with the enabled power
of equational reasoning.

Now to check congruence of R.

-- universally quantified variables
-- i.e., Theorem of (Hidden) Constants
ops s s1 s2 s3 : -> Sender . ops r r1 r2 r3 : -> Receiver .
ops a a1 a2 a3 : -> Abp . op e : -> Nat .

-- a matches a2 while a1 matches a3,
eq a = Protocol(s,r,nil,nil,nil) .

eq a1 = Protocol(s1,r1,nil,nil,nil) .
eq a2 = Protocol(s2,r2,nil,nil,nil) .
eq a3 = Protocol(s3,r3,nil,nil,nil) .

-- relational expansion of R for "a R a2" and "a1 R a3"
eq bit(r) = not bit(s) . eq bit(r2) = not bit(s2) .
eq val(r) = val(r2) . eq val(s) = val(s2) .
eq bit(r1) = bit(s1) . eq bit(r3) = bit(s3) .
eq val(r1) = val(r3) . eq val(s1) = val(s3) .

-- check if R is a behavioural congruence
red In(e,a) R In(e,a2) . -- should be true
red Out(a) R Out(a2) . -- should be true
red In(e,a1) R In(e,a3) . -- should be true
red Out(a1) R Out(a3) . -- should be true

CafeOBJ gives true for all reductions.
We have used the Theorem of (Hidden) Constants to perform proof rules for

universal quantifiers by introducing fresh constant operations. The soundness of
the method is proved in [1, 5]. We have also used implication elimination and case
analysis. One can mimic proof rules of first order logic with equality in a similar
manner. An automated proof support tool called Kumo has been developed for
OBJ3 and CafeOBJ systems [10]. Kumo is not just a proof checking tool, but
also helps to publish proofs over the internet, i.e., Kumo creates webpages for
proofs as it checks proofs given in the proof scripts.

Now that we have established a behavioural congruence, we can use be-
havioural coinduction to prove behavioural equations. The general case is rather
complicated, however, if we restrict attention to ABP states with empty channels
as the representatives of equivalent classes modulo R, the coinduction proof be-
comes manageable using reduction. The following CafeOBJ score does this by
unfolding conditionals. Note that the constants a and a1 are the same as defined
earlier.

var B : Bool
eq not(not(B)) = B . -- necessary! inductive theorem on Bool

red Empty?(init) . -- should be true
red Empty?(Out(a)) == true . -- should be true

red Empty?(In(e,a)) == false . -- should be true
red Empty?(In(e,a1)) == false . -- should be true

red Val(Out(In(e,a1))) == e . -- should be true
red In(e,a) R a . -- should be true
red Out(a1) R a1 . -- should be true

Again, CafeOBJ returns true for all reductions 8. We can now claim that

“ABP implements BUF when there are no errors in channels.”

However, the error recovering capability of ABP is automatic since any loss or du-
plication in the channel does not affect the (now proved) behavioural congruence
of R. Therefore, we arrive at the following correctness theorem for ABP.

Theorem 3. As far as the reachable states are concerned, errors in channels
do not affect the behaviour of the protocol. In other words, the ABP is a robust
implementation of a buffer BUF of capacity one.

Mechanising the whole process should be possible by defining an appropriate
merge operation on queues and the filter that checks bit changes. However, it
is going to be an induction flood. We do not feel it necessary to completely
mechanise proofs when human can do a much better job. We will see a fully
automated verification example using a theorem prover in the next section.

4 Cache Coherence Protocol

In this section, we present a somewhat more implementation-oriented example
of a cache coherence protocol.

A number of computation units share a main memory through a common ac-
cess bus. Each unit issues access requests through the bus and maintains its own
cached copy. In order to assure consistency among cached copies, the protocol
watches (snoops) requests on the bus and takes appropriate actions depending
on the flag values that tracks cache states. A coherence condition we consider is
the following.

“If there are two shared copies, they must be of the same value.”

We model the Illinois cache protocol [11] following [12]. There are three
types of requests, (read, write, and replacement), and four different flag values
(cache states):

– invalid – obsolete;
– shared – not modified, possible copies in other caches;
– valid-exclusive – not modified, only copy in caches;
– dirty – modified, only copy in caches.

Coherence actions are taken as follows, depending on the cache state of the
requesting unit and the existence of other cached copies.

Read Hit. the cache (in the requesting unit) is not invalid, no extra action;
Read Miss. the cache is invalid, look for other cached copies;
8 The double negation equation is necessary. The situation is frequently encountered,

in which coinduction (or invariance proof) on hidden sorts requires induction on
visible sorts.

............Unit#1 Unit#2 Unit#3 Unit#n

Memory

 read
 write
replace

Protocol

Fig. 2. The Illinois Cache Protocol

– dirty cache provides the latest value, both end up shared;
– any other valid-exclusive or shared cache provides the latest value,

all caches having a copy end up shard;
– no cached copy, get a valid-exclusive copy from the main memory;

Write Hit. the cache is not invalid;
– the cache is dirty, no extra action;
– valid-exclusive turns dirty;
– shared turns dirty, all other shared copies are invalid-ated;

Write Miss. invalid turns dirty, all cached copies are invalid-ated;
Replacement. dirty copy is written back to the main memory, stays dirty;

Below is a behavioural specification PROTOCOL of the above procedure.

mod! FLAG {
[ Flag ]
ops invalid valid-exclusive dirty shared : -> Flag

}
mod* PROTOCOL {
protecting(NAT + BOOL + FLAG)
*[ Protocol ]*
bop flag : Nat Protocol -> Flag -- cache state
bop cdata : Nat Protocol -> Nat -- cache value
bop mdata : Protocol -> Nat -- memory value
bop read : Nat Protocol -> Protocol
bop write : Nat Nat Protocol -> Protocol
op init : -> Protocol
vars I J K M N : Nat var P : Protocol

-- initial state
eq flag(I,init) = invalid .



1638 Akira Mori and Kokichi Futatsugi

-- write
eq cdata(I,write(I,M,P)) = M .
ceq cdata(J,write(I,M,P)) = cdata(J,P) if I =/= J .
eq flag(I,write(I,M,P)) = dirty .
-- invalidation
ceq flag(J,write(I,M,P)) = invalid if I =/= J .
ceq mdata(write(I,M,P)) = mdata(P) .

-- read
-- read hit
bceq read(I,P) = P if flag(I,P) =/= invalid .
-- if there is a dirty copy Cj then
eq cdata(I,read(I,write(J,M,P)))=M.--Cj provides missing block
eq mdata(read(I,write(J,M,P))) = mdata(P) .
ceq flag(I,read(I,write(J,M,P))) = shared -- and Ci, Cj

if I =/= J .
ceq flag(J,read(I,write(J,M,P))) = shared -- end up shared.

if I =/= J .
-- if there is a clean copy Cj then
ceq cdata(I,read(I,read(J,P))) = cdata(J,read(J,P)) if I =/= J .

-- Cj provides the missing block
ceq flag(I,read(I,read(J,P)))=shared if I =/= J.--and Ci,Cj
ceq flag(J,read(I,read(J,P)))=shared if I =/= J.--end up shared.
-- independence
beq read(I,read(I,P)) = read(I,P) .
ceq flag(I,read(J,read(K,P))) = flag(I,read(K,P))

if I =/= J and I =/= K .
ceq cdata(I,read(J,P)) = cdata(I,P) if I =/= J .
eq mdata(read(I,P)) = mdata(P) .
-- if there is no cached copy (i.e., only in initial state)
eq cdata(I,read(I,init)) = mdata(init) .
eq flag(I,read(I,init)) = valid-exclusive .
eq mdata(read(I,init)) = mdata(init) .

}

A few remarks on the specification.

– It does not have any non-behavioural operations, thus equational reasoning
is trivially sound.

– Unlike ABP, it handles arbitrary number of units, which is modeled by oper-
ations having an index argument for the unit.

– It only models a single memory block since multiple blocks can be modeled
by an extra argument of each operation.

– It does not model replacement since it has little effect on the coherence
property (and due to the limit of space).

– It is written in a deterministic fashion to allow symbolic execution with term
rewriting (reduction) in CafeOBJ. It is based on the following decision and
observation.

• The unit that made a read request most recently is the one who provides
the value.

• write always gets a dirty copy and invalidates other copies.

As we mentioned earlier, getting inconsistent specifications is much more
likely in behavioural specification. We had great help from a resolution / paramod-
ulation theorem prover that was implemented on CafeOBJ, in finding subtle
inconsistencies among attribute values in the earlier versions.

Now we want to show the following coherence condition.

In all reachable states S:Protocol,
∀I,J:Nat flag(I,S)=flag(J,S)=shared⇒ cdata(I,S)=cdata(J,S) .

Equational reasoning cannot prove this because it cannot trace all reachable
states with respect to behavioural equivalence. The proof requires a technique
similar to coinduction, known as invariance proof. (See [13] for a category-
theoretic characterisation of invariants including relation to bisimulation, and
[14] for a fixed-point characterisation.) It is a well-studied topic in the field of
program verification and is also a prototype of (safety) model checking. There
are two equivalent ways of proving invariance, forward and backward. Backward
calculation of largest invariants (also known as weakest preconditions) is partic-
ularly suitable for behavioural specification. Making a long story short, to prove
that a predicate P (S : h) on a hidden sort h is an invariant for all reachable
states, it suffices to show that the initial state satisfies P and that P is closed
under all methods, that is, ∀S : h ∀X : v P (S) ⇒ P (m(X,S)) for all method
m ∈ Σb

vh,h (e.g., [13]). In our case, the predicate on Protocol is defined as
follows.

P(S:Protocol)=
∀I,J:Nat flag(I,S)=flag(J,S)=shared ⇒ cdata(I,S)=cdata(J,S).

Thus, in order to prove the invariance of P , we have to prove the following
formulas.

P(init).
∀S:Protocol ∀N:Nat P(S) ⇒ P(read(N,S)) .
∀S:Protocol ∀N,M :Nat P(S) ⇒ P(write(N,M,S)) .

It took about five minutes for our resolution/paramodulation theorem prover to
get the proofs of these on a Pentium 266MHz PC 9. The process is completely
automatic except for the inductive theorems on visible sort (in our case, Flag)
mentioned below. The user gives a predicate and the system generates all proof
obligations in clausal form, which are then passed to the prover. Taking five
minutes may not sound great. However, the proof is not easy for a human either.
One of the authors tried the proof and it took more than five minutes. It is
a straight-forward case analysis and we believe that resolution is particularly
9 The first one is trivial, the third one is easy, and the second one took almost the

entire execution time.

useful for this type of theorem proving. Although this proof is probably close
to the practical limit of automated theorem provers, we see many interesting
applications to behavioural specifications.

For example, the above proof requires inductive theorems on the FLAG spec-
ification10.

∀F:Flag (F=invalid ∨ F=valid-exclusive ∨ F=dirty ∨ F=shared) .
invalid 6= valid− exclusive .
invalid 6= dirty .
· · ·

It is an interesting challenge to work with full-fledged inductive theorem provers
so that these inductive theorems are discovered automatically when needed.

The example here may be interesting as a cache coherence proof in its own
right since there are very few examples that are: 1) fully automated (no human
intervention), 2) ready for implementation , 3) free of syntactic translation, and
4) machine supported, despite the large volume of research devoted to the topic.

The invariance proof is too simple when regarded as model checking. How-
ever, by combining iterative calculations of largest fixed-points, one can perform
full-scale model checking. We are currently working on a so-called behavioural
model checker that manages predicate representations, obligation generation,
and theorem prover (resolution + induction) control. This is an interesting area
of study, i.e., the combination of deductive and model-theoretic methods for
software verification.

5 Concluding Remarks

In this paper, we have presented methods for verifying behavioural specifica-
tions based on hidden algebra. We have reported on CafeOBJ’s automation
support with a couple of non-trivial examples, the Alternating Bit Protocol and
a snooping cache coherence protocol. The refinement proof based on behavioural
coinduction is given for the first example, and the coherence proof based on in-
variance is given for the second. We have included almost complete specification
code and proof scores since they are much more valuable and effective sometimes
than verbal descriptions and explanations.

Since the hidden algebra (or behavioural) approach is a relatively new branch
of formal methods, we compare its characteristics with other well-established11

methods.

Process algebra treats observability based on events (rather than values) and
is very suitable for defining process interaction and evolution, which is one
of the unexplored fields of study in hidden algebra. On the other hand,
hidden algebra fits very well the notion of object orientation (methods/
attributes, encapsulated states, inheritance, overloading, and so on). The

10 These were given to the system in advance.
11 Thus some references are omitted.

examples presented in the paper are ready for implementation in Java, for
instance. Also hidden algebra fully inherits the power of abstract data type
specification, by which one can define static, architectural aspects of software
systems (by modules expressions) as well as basic data types within them.

Temporal logic enjoys the rigor of mathematical logics and has made consid-
erable impact on formal verification through model checking technologies.
We are afraid, however, that it may not support software design process
(via refinement) very well since implementing temporal formulas (again for
example in Java) can be very difficult. We are developing techniques for
model checking behavioural specifications, which is briefly touched upon in
Section 4, and planning to use temporal formulas as an input language.

I/O Automata [15] and evolving algebra [16] are probably the closest to
the hidden algebra approach. This is not surprising since hidden algebra
stems from early formalisation efforts for automata. We feel that these tech-
niques are more or less compatible. However, hidden algebra has a clear
relation with ordinary algebra via behavioural equivalence and can rely on
equational reasoning with machine support systems such as CafeOBJ.

Coalgebra [13] stresses duality with algebra and has produced many nice the-
oretical (mostly category-theoretic) results including the construction of ter-
minal coalgebra, Birkhoff like axiomatisability results, and so on. Hidden
algebra (without non-behavioural operations) is actually a concrete case of
coalgebra (e.g.,[17]) just like many-sorted algebra is a concrete case of algebra
(of endo-functors). Coalgebra provides strong supports for hidden algebra in
the area of theoretical development.

We like to mention briefly a couple of related techniques that we think are
effective if used with behavioural specification.

Abstract interpretation [14] can go beyond the limit of refinement-based
verification as it goes in the other direction. However, giving a right abstrac-
tion can be very difficult. Combining these two (refinement and abstraction)
seems very promising. Various safety properties can be proved in this way.

Model checking has become a major topic in automated verification because
of the increased computational capability enabled by binary decision dia-
grams (BDD) techniques. Application is still limited to hardware since the
BDD techniques only apply to finite state systems. It is now a common per-
ception that model checking for software systems requires a combined use of
deductive and model-theoretic techniques. We feel behavioural specification
offers a very good common ground for this. The example in Section 4 is the
first step toward this.

We are building an integrated specification/verification environment on top
of the CafeOBJ system. The planned features include a more efficient built-in
resolution engine, an inductive theorem prover, and a behavioural model check-
ing system. We also plan to have more substantial examples of distributed algo-
rithms, fault-tolerant protocols, and security protocols. This is very important
not just for theoreticians to get feedback, but also for practitioners to enhance
the applicability of the method.

A A CafeOBJ Session for the Alternating Bit Protocol

mori(d194-054)[1024]cafeobj
-- loading standard prelude
Loading /usr/local/cafeobj-1.4/prelude/std.bin
Finished loading /usr/local/cafeobj-1.4/prelude/std.bin

-- CafeOBJ system Version 1.4.2(b3+) --
built: 1998 Nov 28 Sat 6:29:00 GMT

prelude file: std.bin
***

1999 Feb 10 Mon 16:46:26 GMT
Type ? for help

---
uses GCL (GNU Common Lisp)

Licensed under GNU Public Library License
Contains Enhancements by W. Schelter

CafeOBJ> in abp
processing input : ./abp.mod
-- defining module! DATA_*.._* done.
-- defining module! QUEUE_*_*........_....* done.
-- defining module* SENDER........._...*
** system already proved =*= is a congruence of SENDER done.
-- defining module* RECEIVER........._...*
** system already proved =*= is a congruence of RECEIVER done.
-- defining module* ABP,,,,,,,*_*............._.
** system failed to prove =*= is a congruence of ABP done.
CafeOBJ> in test
processing input : ./test.mod
-- opening module ABP.. done.
-- reduce in % : Val(Out(In(2,Out(In(1,Init)))))
2 : NzNat
(0.000 sec for parse, 99 rewrites(0.020 sec), 142 matches)
-- reduce in % : Val(Out(In(2,In(1,Init))))
1 : NzNat
(0.000 sec for parse, 46 rewrites(0.000 sec), 78 matches)
-- reduce in % : Val(Out(In(3,In(2,In(1,Init)))))
1 : NzNat
(0.000 sec for parse, 53 rewrites(0.010 sec), 96 matches)
-- reduce in % : Val(Out(In(3,In(2,Out(In(1,Init))))))
2 : NzNat
(0.000 sec for parse, 114 rewrites(0.010 sec), 170 matches)
-- reduce in % : Val(Out(In(3,Out(Out(In(2,Out(In(1,Init))))))))
3 : NzNat
(0.000 sec for parse, 250 rewrites(0.020 sec), 354 matches)
CafeOBJ>

